IndoTERM: The basic concept, theoretical foundation, and assumption

Day 1 Session 2

Stylized GE model: material flows



Stylized GE model: database table of transaction values

		Producers			Absorption				Total
		Primary	Manufact	Services	H'holds	Invest	Govmt	Export	Demand
Domestic goods	Primary	5	10	2	1	0	0	15	33
	Manufact	5	15	10	15	15	10	10	80
	Services	5	13	20	30	5	20	5	98
Imported goods	Primary	1	2	0	0	0	0		3
	Manufact	1	3	2	3	3	2		14
	Services	1	3	4	6	1	4		19
Primary factors	Labour	7	17	40					
	Capital	8	17	20					
Total	cost	33	80	98					

Simplifications: no taxes, one region only

Production costs = Domestic sales of domestic good

Stylized GE model: database table of transaction values

	Producer i	Absorption C I G	Export	Total Demand
Domestic good c	....	....	....	sum(left)
Imported good c	....	....		
Primary factor f	....			
Production cost	total cost of above			

Value of dom good c used by sector i

Costs = Sales

We need to determine a quantity for each cell, and a price for each row

Stylized GE model: demand equations

	Producer i	Absorption C I G	Export	Total Demand
Domestic good c	$Q_i F(P/PD_c)$	$E_u F(P,PD_c)$	$F(1/PD_c)$	Q_c= sum(left)
Imported good c	$Q_i F(P/PM_c)$	$E_u F(P,PM_c)$		
Primary factor f	$Q_i F(P/PF_f)$			QF_f= sum(left)
Production cost	total cost of above = $PD_i Q_i$			

Supply = demand

Quantity of good c used by sector i

Costs = Sales

Notation:

PD_c = price dom good c	PM_c = price imp good c	PF_f = price factor f	P = full price vector [PD,PM,PF]
Q_i = output good i	F = various functions	QF_f = supply factor f	E_u = expenditure final user u

Stylized CGE model: Number of equations = number endogenous variables

Variable	Determined by:
PD_c = price dom good c	**ZERO PURE PROFITS** values of sales = $PD_c Q_c$ = sum(input costs) = F(all variables)
Q_c = output good c	**MARKET CLEARING** Q_c = sum(individual demands) = F(all variables)
PF_f = price factor f	**For each f, one of PF or QF fixed, the other determined by:**
QF_f = quantity factor f	QF_f = sum(individual demands) = F(all variables)
E_u = spending final user u	**either fixed, or linked to factor incomes (with more equations)**
PM_c = price imp good c	**fixed**

Red: exogenous (set by modeler)

Green: endogenous (explained by system)

IndoTERM

IndoTERM, contains:

- 39 sectors,
- 30 regions,
- 4 labour types,
- 1 household type

Coefficients and Variables

Coefficients

example: USE(c,s,u,d)   UPPER CASE

Mostly values

Either read from file
  or computed with formulae

Constant during each step

Variables

example: puse (c,s,d)   lower case

Often prices or quantities

Percent or ordinary change

Related via equations

Exogenous or endogenous

Vary during each step

The TERM Naming System

or *GLOSS*

COEFFICIENT

variable

Name Part

cap capital
lab labour
lnd land
prim all primary factors

mar margins
tax indirect taxes
pur at purchasers' prices
imp imports (duty paid)
tot total inputs for a user

Prefix

none for levels flow

p % price

x % quantity

del ord.change

LAB(i,o,d)

plab_o(i,d)

xlab_id(o)

Subscripts

c COMmodities
s SouRCe (dom/imp)
i INDustries
m MARgin
o OCCupation
d,r,p Region

Underscore **means "adding up over"**

Core Data and Variables

We begin by declaring variables and data coefficients which appear in many different equations.

Other variables and coefficients will be declared as needed.



TERM Database structure

file data

addups

Index Set		Description
c	COM	Commodities
s	SRC	Domestic or imported (ROW) sources
m	MAR	Margin commodities
r	ORG	Regions of origin
d	DST	Regions of use (destination)
p	PRD	Regions of margin production
f	FINDEM	Final demanders(HOU, INV,GOV, EXP)
i	IND	Industries
u	USR	Users = IND + FINDEM
o	OCC	Skills
h	HOU	Households

PRIMARY FACTOR USE

Inputs to production: Nests



Nested Structure of production

In each industry: Output = function of inputs:

output = F(inputs) = F(Labour, Capital, Land, dom goods, imp goods)

Separability assumptions simplify the production structure:

output = F(primary factor composite, composite goods)

where:

primary factor composite = CES(Labour, Capital, Land)

labour = CES(Various skill grades)

composite good (i) = CES(domestic good (i), imported good (i))

All industries share common production structure.

BUT: Input proportions and behavioural parameters vary.

Nesting is like staged decisions:

First decide how much leather to use—based on output.

Then decide import/domestic proportions, depending on the relative prices of local and foreign leather.

Each nest requires 2 or 3 equations.

Skill Mix


Boxes show
VALUE
price %
quantity %
Labour
LAB_O(i,d)
plab_o(i,d)
xlab_o(i,d)
CES
0.35
Labour type 1
Labour type 2
up to
Labour type O
LAB(i,o,d)
plab(i,o,d)
xlab(i,o,d)

Skill Mix

Problem: for each industry i, choose labour inputs XLAB(i,o,d)

to minimize labour cost:

*sum{o,OCC, PLAB(i,o,d)*XLAB(i,o,d)}*

such that XLAB_O(i) = CES(All,o,OCC: XLAB(i,o,d))

given

Coefficient

(all,i,IND) SIGMALAB(i) # CES substitution between skills #;

(all,i,IND)(all,d,DST) LAB_O(i,d) # Total labour bill in industry i #;

Read SIGMALAB from file INFILE header "SLAB";

Formula (all,i,IND)(all,d,DST)

LAB_O(i,d) = sum{o,OCC, LAB(i,o,d)};

add over OCC

CES Skill Substitution



The problem of zeroes: ID01 function

E_plab_o # Price to each industry of labour composite #

(all,i,IND) (all,d,DST)

LAB_O(i,d)*plab_o(i,d) = sum{o,OCC, LAB(i,o,d)*plab(i,o,d)};

same as

plab_o(i,d) = sum{o,OCC, [LAB(i,o)/LAB_O(i)]* plab(i,o)};

Skill share

What if industry used no labour -- a problem !

ID01(X) = 1 if X=0 otherwise = X

ID01[LAB_O(i,d)]*plab_o(i,d) = sum{o,OCC, LAB(i,o,d)*plab(i,o,d)};

if no labour gives: plab_o(i,d) = 0 (satisfactory)

otherwise gives:

LAB_O(i,d)*plab_o(i,d) = sum{o,OCC, LAB(i,o,d)*plab(i,o,d)};

Primary factor Mix



Primary factor Mix

XPRIM(i,d) = CES(XLAB_O(i,d)/ALAB_O(i,d),
XCAP(i,d)/ACAP(i,d),
XLND(i,d)/ALND(i,d))

quantity-augmenting technical change

Equation
E_xlab_o *# Industry demands for effective labour #*
(all,i,IND)(all,d,DST) xlab_o(i,d) - alab_o(i,d) =
xprim(i,d) - SIGMAPRIM(i)*[plab_o(i,d) + alab_o(i,d) - pprim(i,d)];

E_pcap *# Industry demands for capital #*
(all,i,IND)(all,d,DST) xcap(i,d) - acap(i,d) =
xprim(i,d) - SIGMAPRIM(i)*[pcap(i,d) + acap(i,d) - pprim(i,d)];

E_plnd *# Industry demands for land #*
(all,i,IND)(all,d,DST) xlnd(i,d) - alnd(i,d) =
xprim(i,d) - SIGMAPRIM(i)*[plnd(i,d) + alnd(i,d) - pprim(i,d)];

E_pprim *# Effective price term for factor demand equations #*
(all,i,IND)(all,d,DST)
PRIM(i,d)*pprim(i,d) = LAB_O(i,d)*[plab_o(i,d) + alab_o(i,d)]
+ CAP(i,d)*[pcap(i,d) + acap(i,d)] + LND(i,d)*[plnd(i,d) + alnd(i,d)];

Intermediate Sourcing



PUR_S(c,i,d)
ppur_s(c,i,d)
xint_s(c,i,d)

Boxes show
VALUE
price %
quantity %

PUR(c,s,i,d)
ppur(c,s,i,d)
xint(c,s,i,d)

SAME FOR HOU AND INV USERS

Import/Domestic Substitution



INTERMEDIATE USE

Intermediate Sourcing

Variable
(all,c,COM)(all,u,USR)(all,d,DST) ppur_s(c,u,d) *# User prices, average over s#*;
(all,c,COM)(all,i,IND)(all,d,DST)
xint_s(c,i,d) *# Industry demands for dom/imp composite #*;
Equation
E_ppur_s
(all,c,COM)(all,u,USR)(all,d,DST)
ppur_s(c,u,d) = sum{s,SRC,SRCSHR(c,s,u,d)*ppur(c,s,u,d)};

E_xint (all,c,COM)(all,s,SRC)(all,i,IND)(all,d,DST)
xint(c,s,i,d) = xint_s(c,i,d) - SIGMADOMIMP(c)*[ppur(c,s,i,d)-ppur_s(c,i,d)];

$$x_s = x_{average} - s[p_s - p_{average}]$$

$$p_{average} = \sum S_s.[p_s]$$

SAME FOR HOU AND INV USERS

Numerical Example of CES demands

feel for numbers

$p = S_d p_d + S_m p_m$ **average price of dom and imp Food**

$x_d = x - \sigma(p_d - p)$ **demand for domestic Food**

$x_m = x - \sigma(p_m - p)$ **demand for imported Food**

Let p_m=-10%, $x=p_d=0$

Let S_m=0.3 and σ =2. This gives:

$p = -0.3*10 = -3$

$x_d = - 2(- -3) = -6$ $x_d = x - S_m\sigma(p_d - p_m)$

$x_m = -2(-10 - - 3) = 14$ $x_m = x - S_d\sigma(p_m - p_d)$

Cheaper imports cause 14% increase in import volumes and 6% fall in domestic demand.

Effect on domestic sales is proportional to both S_m and σ.

THE TRADE SYSTEM

TRADE(“machi","dom")

User or destination region

Producer or source region

TRADE	1 NAD	2 SUMUT	3 SUMBAR	4 RIAU	5 JAMBI	6 SUMSEL	7 BABEL	8 BENGKULU	9 LAMPUNG	10 DKI
1 NAD	0,053	0,002	0,000	0,000	0,000	0,000	0,000	0,000	0,000	0,000
2 SUMUT	301	740	172	2,15	31,2	108	2,70	16,9	53,1	0,913
3 SUMBAR	0,002	0,001	0,038	0,000	0,000	0,002	0,000	0,000	0,001	0,000
4 RIAU	252	1097	312	13076	254	676	36,7	66,3	252	74,9
5 JAMBI	0,000	0,000	0,001	0,000	0,013	0,007	0,000	0,000	0,001	0,000
6 SUMSEL	0,001	0,001	0,000	0,000	0,001	0,076	0,000	0,000	0,005	0,000
7 BABEL	0,000	0,000	0,000	0,000	0,000	0,001	0,005	0,000	0,000	0,000
8 BENGKULU	0,000	0,000	0,001	0,000	0,000	0,003	0,000	0,006	0,001	0,000
9 LAMPUNG	0,000	0,000	0,000	0,000	0,000	0,002	0,000	0,000	0,036	0,000
10 DKI	255	798	312	36,1	90,5	344	83,2	49,7	295	9406
11 JABAR	2650	8387	2963	3709	912	3532	675	501	2917	42782
12 BANTEN	32,6	103	37,1	45,7	11,7	46,8	10,3	6,29	35,8	962
13 JATENG	18,5	39,5	17,5	0,552	5,82	23,5	2,99	3,19	18,7	3,05
14 DIY	3,39	9,58	3,08	0,874	1,06	4,29	0,516	0,582	3,34	3,92
15 JATIM	32,3	34,6	27,5	0,991	10,5	43,4	3,99	5,28	30,8	3,04
16 KALBAR	0,001	0,000	0,000	0,000	0,000	0,001	0,000	0,000	0,001	0,000

One matrix for each domestic commodity and each imported.
Diagonal shows goods produced and used in same region.
Row and column totals given by USE matrix.
Otherwise, made up by gravity and other assumptions.

TRADE("machi","imp")

User or destination region

Arrival port region

Special View Menu

TRADE	22 SULTENG	23 SULSEL	24 SULTRA	25 BALI	26 NTB	27 NTT	28 MALUKU	29 MALUT	30 PAPUA	Total
1 NAD	1,03	0,009	1,76	2,14	2,82	1,35	0,587	0,231	8,81	1402
2 SUMUT	7,21	0,283	12,8	15,6	20,5	9,33	4,09	1,53	55,2	5508
3 SUMBAR	1,07	0,278	1,95	2,89	3,71	1,59	0,612	0,222	8,24	1622
4 RIAU	5,92	0,248	10,4	12,8	16,1	7,19	2,99	1,10	38,6	8835
5 JAMBI	0,175	0,001	0,286	0,390	0,432	0,239	0,075	0,039	1,38	137
6 SUMSEL	1,01	0,005	1,53	2,32	2,36	1,46	0,350	0,241	8,97	916
7 BABEL	0,232	0,001	0,393	0,563	0,652	0,282	0,103	0,041	1,44	147
8 BENGKULU	0,032	0,000	0,048	0,083	0,079	0,049	0,010	0,008	0,285	18,8
9 LAMPUNG	2,88	0,860	5,17	7,27	9,56	4,25	1,70	0,618	23,4	2310
10 DKI	6,72	3,26	13,6	29,1	32,4	11,3	3,62	1,19	44,6	41232
11 JABAR	0,037	0,004	0,071	3,36	0,182	1,21	0,019	0,122	4,5[illegible]	355
12 BANTEN	0,500	0,003	0,978	2,09	2,30	0,846	0,259	0,091	[illegible],47	1825
13 JATENG	11,7	5,99	24,2	82,2	75,9	21,1	5,74	1,82	67,3	14611
14 DIY	0	0	0	0	0	0	0	0	0	0
15 JATIM	18,7	1,24	39,8	253	172	36,8	8,55	2,65	96,5	13862
16 KALBAR	0,881	0,005	1,41	2,42	2,36	1,10	0,297	0,1[illegible]7	4,96	226

provinces have big ports

Market clearing for domestic goods

sum over IND: **MAKE** (COM,IND,ORG) = **MAKE_I** (COM,ORG) = **TRADE_D** (COM,ORG) = sum over DST: **TRADE** (COM,ORG,DST)

```
E_xtrad_d (all,c,COM)(all,s,SRC)(all,r,ORG)
TRADE_D(c,s,r)*xtrad_d(c,s,r)
   = sum{d,DST, TRADE(c,s,r,d)*xtrad(c,s,r,d)};

Equation E_pdomA # Demand = supply for non-margins #
 (all,c,NONMAR)(all,r,REG) xcom(c,r) = xtrad_d(c,"dom",r);

Equation E_pdomB # Demand = supply for margins #
 (all,m,MAR)(all,p,REG) MAKE_I(m,p)*xcom(m,p) =
 TRADE_D(m,"dom",p)*xtrad_d(m,"dom",p) + SUPPMAR_RD(m,p)*xsuppmar_rd(m,p);
```

HOUSEHOLD DEMAND



HOU column in USE is driven by adding up (over HOU) the HOUPUR matrix

header 3PUR

Equation E_xhou_s (all,c,COM)(all,d,DST)
xhou_s(c,d) = sum{h,HOU, HOUSHR(c,h,d)*xhouh_s(c,h,d)};

Top Nest of Household Demands



Klein-Rubin:
a non-homothetic utility function

Homothetic means:

budget shares depend only on prices, not incomes
eg: CES, Cobb-Douglas

Non-homothetic means:

rising income causes budget shares to change
even with price ratios fixed.

Non-unitary expenditure elasticities:
I% rise in total expenditure might cause food expenditure to rise by 1/2%; air travel expenditure to rise by 2%.

Other names: Stone-Geary
or LES: linear expenditure system

Linear Expenditure System

Total expenditure = subsistence cost + luxury expenditure

supernumerary

PHOU(c) *XHOU(c) = PHOU(c) *XSUB(c) + SLUX(c) *LUX_C

PHOU(c) *XHOU(c) = PHOU(c) *XSUB(c)

+ SLUX(c) *[HOU_C - Σ {PHOU(c) *XSUB(c)}]

all subsistence costs

Expenditure on each good

is a linear function of prices and income

Model demand equations

Total demand = subsistence demand + luxury demand

Equation

E_xsub *# Subsistence demand for composite commodities #*

(all,c,COM)(all,h,HOU)(all,d,DST) xsub(c,h,d) = nhouh(h,d) + asub(c,h,d);

E_xlux *# Luxury demand for composite commodities #*

(all,c,COM)(all,h,HOU)(all,d,DST)

xlux(c,h,d) + phou(c,d) = wlux(h,d) + alux(c,h,d);

E_xhouh_s *# Total household demand for composite commodities #*

(all,c,COM)(all,h,HOU)(all,d,DST)

xhouh_s(c,h,d) = BLUX(c,h,d)*xlux(c,h,d) + [1-BLUX(c,h,d)]*xsub(c,h,d);

How many parameters -degree of flexibility

No of parameters =

extra numbers needed to specify percent change form

IF EXPENDITURE VALUES ARE ALREADY KNOWN

Example, CES=1:

with input values known, 1 number, σ, is enough.

Example, CobbDouglas=0:

with input values known, we know all.

In levels, more parameters are needed.

Example, Leontief=0:

with input values known, we know all.

How many parameters is Klein-Rubin/LES ?

We need to divide expenditure on each good

into subsistence and luxury parts.

(all,c,COM) BLUX(c) # Ratio,supernumerary/total expenditure#;

One BLUX parameter for each commodity.

These "parameters" change !

INVENTORY DEMAND



INV column in USE is driven by adding up (over IND) the INVEST matrix

header 2PUR

E_xinv_s (all,c,COM)(all,d,DST)
INVEST_I(c,d)*xinv_s(c,d)= sum{i,IND, INVEST(c,i,d)*xinvi(c,i,d)};

Investment Composition differs by Industry

COM \ IND	....	Manufact	....	Estate	Service
....					
....					
....					
Equipm					
....					
Constr					
....					
Service					

Machines for industry

New houses

Composition of Investment

Each industry has its own fixed recipe to make its new capital.



Equation E_xinvi (all,c,COM)(all,i,IND)(all,d,DST)
xinvi(c,i,d) = xinvitot(i,d);

Driving industry Investment

```
Variable
 (all,i,IND)(all,d,DST)  gret(i,d)
   # Gross rate of return = Rental/[Price of new capital] #;
 (all,i,IND)(all,d,DST)  ggro(i,d)
   # Gross growth rate of capital = Investment/capital #;
 (all,i,IND)(all,d,DST)  finv1(i,d)  # Investment shift variable #;
 invslack # Investment slack variable for exogenizing national investment #;

Equation
 E_gret (all,i,IND)(all,d,DST)       gret(i,d)  = pcap(i,d) - pinvitot(i,d);
 E_xinvitot (all,i,IND)(all,d,DST)  ggro(i,d) = xinvitot(i,d) - xcap(i,d);

Equation E_ggro # DPSV investment rule #
 (all,i,IND)(all,d,DST) ggro(i,d) = finv1(i,d) + 0.33*[2.0*gret(i,d) -invslack];
```

INVESTMENT/CAPITAL ratio GGRO is either

- fixed -- in long run, or
- related to sector profits

Longrun Investment Rule

Investment/capital ratios GGRO are exogenous

E_xinvitot (all,i,IND)(all,d,DST) ggro(i,d) = xinvitot(i,d) - xcap(i,d);



GOVERNMENT & EXPORT DEMAND

Government demands

Equation E_xgov (all,c,COM)(all,s,SRC)(all,d,DST)
xgov(c,s,d) = fgovtot(d) + fgov(c,s,d) + fgov_s(c,d) + fgovgen;

Equation E_fgovtot2 (all,d,REG)
fgovtot(d) = fgovtot2(d) + MainMacro("RealHou",d);

Shift variables fgovtot and fgovtot2 used to switch between two rules. With fgovtot2 exogenous, fgovtot endogenous, we get

xgov(c,s,d) = MainMacro("RealHou",d) + fgovtot2(d) + fgov(c,s,d) + fgov_s(c,d) + fgovgen;

ie: gov. demands follow real household consumption

with fgovtot exogenous, fgovtot2 endogenous, we get

xgov(c,s,d) = fgovtot(d) + fgov(c,s,d) + fgov_s(c,d) + fgovgen;
ie: gov. demands are exogenous

Export demands

```
Coefficient (all,c,COM) EXP_ELAST(c)
  # Export demand elasticities: typical value 5.0 #;
Variable
(all,c,COM)(all,s,SRC) fqexp(c,s) # Export quantity shift variable #;
(all,c,COM)(all,s,SRC) fpexp(c,s)    # Export price shift variable #;
(all,c,COM)(all,d,DST) xexp_s(c,d)  # Export demands, dom+imp #;

Equation E_xexp (all,c,COM)(all,s,SRC)(all,d,DST)
   xexp(c,s,d) = fqexp(c,s) - EXP_ELAST(c)
           * [ppur(c,s,"Exp",d) -fpexp(c,s) -phi];
```

MARKET CLEARING

! Total demand for commodity c produced in r = supply commodity c produced in r!

```
Coefficient
(all,c,COM)(all,s,SRC)(all,r,REG) TRDIAG(c,s,r) # Trade matrix diagonal #;
(all,c,COM)(all,s,SRC)(all,r,ORG) TRADE_D(c,s,r) # Total direct demands #;
(all,c,COM)(all,s,SRC)(all,d,DST) TRADE_R(c,s,d) # Total direct demands #;
(all,c,COM)(all,s,SRC)         TRADE_RD(c,s) # Total national direct demands #;

Formula
(all,c,COM)(all,s,SRC)(all,r,REG) TRDIAG(c,s,r)  = TRADE(c,s,r,r);
(all,c,COM)(all,s,SRC)(all,r,ORG) TRADE_D(c,s,r) = sum{d,DST, TRADE(c,s,r,d)};
(all,c,COM)(all,s,SRC)(all,d,DST) TRADE_R(c,s,d) = sum{r,ORG, TRADE(c,s,r,d)};
(all,c,COM)(all,s,SRC)          TRADE_RD(c,s)  = sum{r,ORG, TRADE_D(c,s,r)};
```

```
Variable
    (all,c,COM)(all,s,SRC)(all,r,ORG) xtrad_d(c,s,r)
     # Total direct demands for goods produced(dom) or landed(imp) in r #;
    Equation E_xtrad_d
    (all,c,COM)(all,s,SRC)(all,r,ORG) ID01(TRADE_D(c,s,r))*xtrad_d(c,s,r) =
                          sum{d,DST, TRADE(c,s,r,d)*xtrad(c,s,r,d)};

    Equation E_xtrad_r
    (all,c,COM)(all,s,SRC)(all,d,DST) ID01(TRADE_R(c,s,d))*xtrad_r(c,s,d) =
                          sum{r,ORG, TRADE(c,s,r,d)*xtrad(c,s,r,d)};

    Equation E_pdomA # Supply = demand for non-margins #
     (all,c,NONMAR)(all,r,REG)
      xcom(c,r) = xtrad_d(c,"dom",r);

    Equation E_pdomB # Demand = supply for margins #
     (all,m,MAR)(all,p,REG)
      MAKE_I(m,p)*xcom(m,p) = TRADE_D(m,"dom",p)*xtrad_d(m,"dom",p)
                       + SUPPMAR_RD(m,p)*xsuppmar_rd(m,p);
```